

Unipot

Computing with elements of unipotent subgroups of Chevalley groups

1.7

16 August 2026

Sergei Haller

Sergei Haller Email: sergei@sergei-haller.de

Homepage: <https://www.sergei-haller.de>

Contents

1	Preface	3
1.1	Root Systems	3
1.2	Citing Unipot	3
2	The GAP Package Unipot	4
2.1	General functionality	4
2.2	Unipotent subgroups of Chevalley groups	4
2.3	Elements of unipotent subgroups of Chevalley groups	7
2.4	Symbolic computation	15
	References	16
	Index	17

Chapter 1

Preface

Unipot is a package for GAP 4 [GAP04]. The version 1.0 of this package was the content of my diploma thesis [Hal00].

Let U be a unipotent subgroup of a Chevalley group of type $L(K)$. Then it is generated by the elements $x_r(t)$ for all $r \in \Phi^+, t \in K$. The roots of the underlying root system Φ are ordered according to the height function. Each element of U is a product of the root elements $x_r(t)$. By Theorem 5.3.3 from [Car89] each element of U can be uniquely written as a product of root elements with roots in increasing order. This unique form is called the canonical form.

The main purpose of this package is to compute the canonical form of an element of the group U . To this end we have implemented the unipotent subgroups of Chevalley groups and their elements as GAP objects and installed some operations for them. One method for the operation `Comm` uses Chevalley's commutator formula, which we have implemented, too.

1.1 Root Systems

We use the root systems and the structure constants available in GAP from the simple Lie algebras, and the same ordering of roots.

1.2 Citing Unipot

If you use Unipot to solve a problem or publish some result that was partly obtained using Unipot, I would appreciate it if you would cite Unipot, just as you would cite another paper that you used. (Below is a sample citation.) Again I would appreciate if you could inform me about such a paper.

Specifically, please refer to:

Example
[Hal02] Sergei Haller. Unipot --- a system for computing with elements of unipotent subgroups of Chevalley groups, July 2002.

Chapter 2

The GAP Package Unipot

This chapter describes the package `Unipot`. Mainly, the package provides the ability to compute with elements of unipotent subgroups of Chevalley groups, but also some properties of these groups.

In this chapter we will refer to unipotent subgroups of Chevalley groups as “unipotent subgroups” and to elements of unipotent subgroups as “unipotent elements”. Specifically, we only consider unipotent subgroups generated by all positive root elements.

2.1 General functionality

In this section we will describe the general functionality provided by this package.

2.1.1 `UnipotChevInfo`

▷ `UnipotChevInfo` (info class)

`UnipotChevInfo` is an `InfoClass` used in this package. `InfoLevel` of this `InfoClass` is set to 1 by default and can be changed to any level by `SetInfoLevel( UnipotChevInfo, n )`.

The following levels are used throughout the package:

1. nothing is printed.
2. When calculating the order of a finite unipotent subgroup, the power presentation of this number is printed. (See `Size` (2.2.5) for an example.)
3. When comparing unipotent elements, report for which of them the canonical form must be computed. (See `\=` (2.3.11) for an example.)
4. nothing is printed.
5. While calculating the canonical form, the different steps are printed.
6. The process of calculating the Chevalley commutator constants is printed.

2.2 Unipotent subgroups of Chevalley groups

In this section we will describe the functionality for unipotent subgroups provided by this package.

2.2.1 IsUnipotChevSubGr

▷ IsUnipotChevSubGr(*grp*) (Category)

Category for unipotent subgroups.

2.2.2 UnipotChevSubGr

▷ UnipotChevSubGr(*type*, *n*, *F*) (function)

UnipotChevSubGr returns the unipotent subgroup U of the Chevalley group of type *type*, rank *n* over the ring *F*.

type must be one of "A", "B", "C", "D", "E", "F", "G".

For the type "A", *n* must be a positive integer.

For the types "B" and "C", *n* must be a positive integer ≥ 2 .

For the type "D", *n* must be a positive integer ≥ 4 .

For the type "E", *n* must be one of 6, 7, 8.

For the type "F", *n* must be 4.

For the type "G", *n* must be 2.

Example

```
gap> U_G2 := UnipotChevSubGr("G", 2, Rationals);
<Unipotent subgroup of a Chevalley group of type G2 over Rationals>
gap> IsUnipotChevSubGr(U_G2);
true
```

Example

```
gap> UnipotChevSubGr("E", 3, Rationals);
Error, <n> must be one of 6, 7, 8 for type E called from
UnipotChevFamily( type, n, F ) called from
<function>( <arguments> ) called from read-eval-loop
Entering break read-eval-print loop ...
you can 'quit;' to quit to outer loop, or
you can 'return;' to continue
brk>
```

2.2.3 PrintObj (for UnipotChevSubGr)

▷ PrintObj(*U*) (method)

▷ ViewObj(*U*) (method)

Special methods for unipotent subgroups. (see GAP Reference Manual, section (**Reference: View and Print**) for general information on View and Print)

Example

```
gap> Print(U_G2);
UnipotChevSubGr( "G", 2, Rationals )gap> View(U_G2);
<Unipotent subgroup of a Chevalley group of type G2 over Rationals>gap>
```

2.2.4 One (for UnipotChevSubGr)

- ▷ `One(U)` (method)
- ▷ `OneOp(U)` (method)

Special methods for unipotent subgroups. Return the identity element of the group U . The returned element has representation UNIPOT_DEFAULT_REP (see UNIPOT_DEFAULT_REP (2.3.3)).

2.2.5 Size (for UnipotChevSubGr)

- ▷ `Size(U)` (method)

`Size` returns the order of a unipotent subgroup. This is a special method for unipotent subgroups using the result in Carter [Car89], Theorem 5.3.3 (ii).

Example

```
gap> SetInfoLevel( UnipotChevInfo, 2 );
gap> Size( UnipotChevSubGr("E", 8, GF(7)) );
#I The order of this group is 7^120 which is
258086210989349276047917817413172383631691140276099547911280598425927853437317\
437263620645695945672001
gap> SetInfoLevel( UnipotChevInfo, 1 );
```

2.2.6 RootSystem (for UnipotChevSubGr)

- ▷ `RootSystem(U)` (method)

This method is similar to the method `RootSystem` for semisimple Lie algebras (see Section (Reference: Semisimple Lie Algebras and Root Systems) in the GAP Reference Manual for further information).

`RootSystem` returns the underlying root system of the unipotent subgroup U . The returned object is from the category `IsRootSystem`:

Example

```
gap> R_G2 := RootSystem(U_G2);
<root system of rank 2>
gap> IsRootSystem(last);
true
gap> SimpleSystem(R_G2);
[ [ 2, -1 ], [ -3, 2 ] ]
```

In addition to the properties and attributes described in the Reference Manual, the following attributes are installed for the Root Systems by the package `Unipot`:

2.2.7 PositiveRootsFC

- ▷ `PositiveRootsFC(R)` (attribute)
- ▷ `NegativeRootsFC(R)` (attribute)

The list of positive resp. negative roots of the root system R . Every root is represented as a list of coefficients of the linear combination in fundamental roots. E.g. let $r = \sum_{i=1}^l k_i r_i$, where $r_1, \dots, r_l$ are the fundamental roots, then r is represented as the list $[k_1, \dots, k_l]$.

Example

```

gap> U_E6 := UnipotChevSubGr("E",6,GF(2));
<Unipotent subgroup of a Chevalley group of type E6 over GF(2)>
gap> R_E6 := RootSystem(U_E6);
<root system of rank 6>
gap> PositiveRoots(R_E6){[1..6]};
[ [ 2, 0, -1, 0, 0, 0 ], [ 0, 2, 0, -1, 0, 0 ], [ -1, 0, 2, -1, 0, 0 ],
  [ 0, -1, -1, 2, -1, 0 ], [ 0, 0, 0, -1, 2, -1 ], [ 0, 0, 0, 0, -1, 2 ] ]
gap> PositiveRootsFC(R_E6){[1..6]};
[ [ 1, 0, 0, 0, 0, 0 ], [ 0, 1, 0, 0, 0, 0 ], [ 0, 0, 1, 0, 0, 0 ],
  [ 0, 0, 0, 1, 0, 0 ], [ 0, 0, 0, 0, 1, 0 ], [ 0, 0, 0, 0, 0, 1 ] ]
gap> PositiveRootsFC(R_E6)[Length(PositiveRootsFC(R_E6))]; # the highest root
[ 1, 2, 2, 3, 2, 1 ]

```

2.2.8 GeneratorsOfGroup (for UnipotChevSubGr)

▷ GeneratorsOfGroup(U)

(method)

This is a special Method for unipotent subgroups of finite Chevalley groups.

2.2.9 Representative

▷ Representative(U)

(method)

This method returns an element of the unipotent subgroup U with indeterminates instead of ring elements. Such an element could be used for symbolic computations (see 2.4). The returned element has representation UNIPOT_DEFAULT_REP (see UNIPOT_DEFAULT_REP (2.3.3)).

Example

```

gap> Representative(U_G2);
x_{1}( t_1 ) * x_{2}( t_2 ) * x_{3}( t_3 ) * x_{4}( t_4 ) * x_{5}( t_5 ) * x_{\
6}( t_6 )

```

2.2.10 CentralElement

▷ CentralElement(U)

(attribute)

This method returns the representative of the center of U without calculating the center.

2.3 Elements of unipotent subgroups of Chevalley groups

In this section we will describe the functionality for unipotent elements provided by this package.

2.3.1 IsUnipotChevElem

▷ IsUnipotChevElem(elm)

(Category)

Category for elements of a unipotent subgroup.

2.3.2 IsUnipotChevRepByRootNumbers

- ▷ `IsUnipotChevRepByRootNumbers(elm)` (Representation)
- ▷ `IsUnipotChevRepByFundamentalCoeffs(elm)` (Representation)
- ▷ `IsUnipotChevRepByRoots(elm)` (Representation)

`IsUnipotChevRepByRootNumbers`, `IsUnipotChevRepByFundamentalCoeffs` and `IsUnipotChevRepByRoots` are different representations for unipotent elements.

Roots of elements with representation `IsUnipotChevRepByRootNumbers` are represented by their numbers (positions) in `PositiveRoots(RootSystem(U))`.

Roots of elements with representation `IsUnipotChevRepByFundamentalCoeffs` are represented by elements of `PositiveRootsFC(RootSystem(U))`.

Roots of elements with representation `IsUnipotChevRepByRoots` are represented by roots themselves, i.e. elements of `PositiveRoots(RootSystem(U))`.

(See `UnipotChevElemByRootNumbers` (2.3.4), `UnipotChevElemByFundamentalCoeffs` (2.3.5) and `UnipotChevElemByRoots` (2.3.6) for examples.)

2.3.3 UNIPOT_DEFAULT_REP

- ▷ `UNIPOT_DEFAULT_REP` (global variable)

This variable contains the default representation for newly created elements, e.g. created by `One` or `Random`. When `Unipot` is loaded, the default representation is `IsUnipotChevRepByRootNumbers` and can be changed by assigning a new value to `UNIPOT_DEFAULT_REP`.

Example

```
gap> UNIPOT_DEFAULT_REP := IsUnipotChevRepByFundamentalCoeffs;;
```

Note that `Unipot` doesn't check the type of this value, i.e. you may assign any value to `UNIPOT_DEFAULT_REP`, which may result in errors in the following commands:

Example

```
gap> UNIPOT_DEFAULT_REP := 3;;
gap> One( U_G2 );
... Error message ...
```

2.3.4 UnipotChevElemByRootNumbers

- ▷ `UnipotChevElemByRootNumbers(U, roots, felems)` (operation)
- ▷ `UnipotChevElemByRootNumbers(U, root, felem)` (operation)
- ▷ `UnipotChevElemByRN(U, roots, felems)` (operation)
- ▷ `UnipotChevElemByRN(U, root, felem)` (operation)

`UnipotChevElemByRootNumbers` returns an element of a unipotent subgroup U with representation `IsUnipotChevRepByRootNumbers` (see `IsUnipotChevRepByRootNumbers` (2.3.2)).

`roots` should be a list of root numbers, i.e. integers from the range 1, ..., `Length(PositiveRoots(RootSystem(U)))`. And `felems` a list of corresponding ring elements or indeterminates over that ring (see **GAP Reference Manual**, Indeterminate (**Reference: Indeterminate**) for general information on indeterminates or section 2.4 of this manual for examples).

The second variant of `UnipotChevElemByRootNumbers` is an abbreviation for the first one if `roots` and `felems` contain only one element.

`UnipotChevElemByRN` is just a synonym for `UnipotChevElemByRootNumbers`.

Example

```
gap> IsIdenticalObj( UnipotChevElemByRN, UnipotChevElemByRootNumbers );
true
gap> y := UnipotChevElemByRootNumbers(U_G2, [1,5], [2,7] );
x_{1}( 2 ) * x_{5}( 7 )
gap> x := UnipotChevElemByRootNumbers(U_G2, 1, 2);
x_{1}( 2 )
```

In this example we create two elements: $x_{r_1}(2) \cdot x_{r_5}(7)$ and $x_{r_1}(2)$, where $r_i, i = 1, \dots, 6$ are the positive roots in `PositiveRoots(RootSystem(U))` and $x_{r_i}(t), i = 1, \dots, 6$ the corresponding root elements.

2.3.5 UnipotChevElemByFundamentalCoeffs

- ▷ `UnipotChevElemByFundamentalCoeffs(U, roots, felems)` (operation)
- ▷ `UnipotChevElemByFundamentalCoeffs(U, root, felem)` (operation)
- ▷ `UnipotChevElemByFC(U, roots, felems)` (operation)
- ▷ `UnipotChevElemByFC(U, root, felem)` (operation)

`UnipotChevElemByFundamentalCoeffs` returns an element of a unipotent subgroup U with representation `IsUnipotChevRepByFundamentalCoeffs` (see `IsUnipotChevRepByFundamentalCoeffs (2.3.2)`).

`roots` should be a list of elements of `PositiveRootsFC(RootSystem(U))`. And `felems` a list of corresponding ring elements or indeterminates over that ring (see **GAP Reference Manual, Indeterminate (Reference: Indeterminate)** for general information on indeterminates or section 2.4 of this manual for examples).

The second variant of `UnipotChevElemByFundamentalCoeffs` is an abbreviation for the first one if `roots` and `felems` contain only one element.

`UnipotChevElemByFC` is just a synonym for `UnipotChevElemByFundamentalCoeffs`.

Example

```
gap> PositiveRootsFC(RootSystem(U_G2));
[ [ 1, 0 ], [ 0, 1 ], [ 1, 1 ], [ 2, 1 ], [ 3, 1 ], [ 3, 2 ] ]
gap> y1 := UnipotChevElemByFundamentalCoeffs( U_G2, [[ 1, 0 ], [ 3, 1 ]], [2,7] );
x_{[ 1, 0 ]}( 2 ) * x_{[ 3, 1 ]}( 7 )
gap> x1 := UnipotChevElemByFundamentalCoeffs( U_G2, [ 1, 0 ], 2 );
x_{[ 1, 0 ]}( 2 )
```

In this example we create the same two elements as in `UnipotChevElemByRootNumbers (2.3.4)`: $x_{[1,0]}(2) \cdot x_{[3,1]}(7)$ and $x_{[1,0]}(2)$, where $[1,0] = 1r_1 + 0r_2 = r_1$ and $[3,1] = 3r_1 + 1r_2 = r_5$ are the first and the fifth positive roots of `PositiveRootsFC(RootSystem(U))` respectively.

2.3.6 UnipotChevElemByRoots

- ▷ `UnipotChevElemByRoots(U, roots, felems)` (operation)
- ▷ `UnipotChevElemByRoots(U, root, felem)` (operation)
- ▷ `UnipotChevElemByR(U, roots, felems)` (operation)

▷ `UnipotChevElemByR(U, root, felem)` (operation)

`UnipotChevElemByRoots` returns an element of a unipotent subgroup U with representation `IsUnipotChevRepByRoots` (see `IsUnipotChevRepByRoots` (2.3.2)).

`roots` should be a list of elements of `PositiveRoots(RootSystem(U))`. And `felems` a list of corresponding ring elements or indeterminates over that ring (see GAP Reference Manual, Indeterminate (**Reference: Indeterminate**) for general information on indeterminates or section 2.4 of this manual for examples).

The second variant of `UnipotChevElemByRoots` is an abbreviation for the first one if `roots` and `felems` contain only one element.

`UnipotChevElemByR` is just a synonym for `UnipotChevElemByRoots`.

Example

```
gap> PositiveRoots(RootSystem(U_G2));
[ [ 2, -1 ], [ -3, 2 ], [ -1, 1 ], [ 1, 0 ], [ 3, -1 ], [ 0, 1 ] ]
gap> y2 := UnipotChevElemByRoots( U_G2, [[ 2, -1 ], [ 3, -1 ]], [2,7] );
x_{[ 2, -1 ]}( 2 ) * x_{[ 3, -1 ]}( 7 )
gap> x2 := UnipotChevElemByRoots( U_G2, [ 2, -1 ], 2 );
x_{[ 2, -1 ]}( 2 )
```

In this example we create again the two elements as in previous examples: $x_{[2,-1]}(2) \cdot x_{[3,-1]}(7)$ and $x_{[2,-1]}(2)$, where $[2,-1] = r_1$ and $[3,-1] = r_5$ are the first and the fifth positive roots of `PositiveRoots(RootSystem( U ))` respectively.

2.3.7 UnipotChevElemByRootNumbers (element conversion)

▷ `UnipotChevElemByRootNumbers(x)` (operation)

▷ `UnipotChevElemByFundamentalCoeffs(x)` (operation)

▷ `UnipotChevElemByRoots(x)` (operation)

These three methods are provided for converting a unipotent element to the respective representation.

If x has already the required representation, then x itself is returned. Otherwise a *new* element with the required representation is generated.

Example

```
gap> x;
x_{1}( 2 )
gap> x1 := UnipotChevElemByFundamentalCoeffs( x );
x_{[ 1, 0 ]}( 2 )
gap> IsIdenticalObj(x, x1); x = x1;
false
true
gap> x2 := UnipotChevElemByFundamentalCoeffs( x1 );
gap> IsIdenticalObj(x1, x2);
true
```

Note: If some attributes of x are known (e.g `Inverse` (see `Inverse` (2.3.15)) or `CanonicalForm` (see `CanonicalForm` (2.3.8))), then they are “converted” to the new representation, too.

Example

```
{\fmark}UnipotChevElemByRootNumbers( <U>, <list> ) 0
{\fmark}UnipotChevElemByRoots( <U>, <list> ) 0
{\fmark}UnipotChevElemByFundamentalCoeffs( <U>, <list> ) 0
```

DEPRECATED These are old versions of `UnipotChevElemByXX` (from Unipot 1.0 and 1.1). They are deprecated now and exist for compatibility only. They may be removed at any time.

2.3.8 CanonicalForm

▷ `CanonicalForm(x)`

(attribute)

`CanonicalForm` returns the canonical form of x . For more information on the canonical form see Carter [Car89], Theorem 5.3.3 (ii). It says:

Each element of a unipotent subgroup U of a Chevalley group with root system Φ is uniquely expressible in the form

$$\prod_{r_i \in \Phi^+} x_{r_i}(t_i),$$

where the product is taken over all positive roots in increasing order.

Example

```
gap> z := UnipotChevElemByFC( U_G2, [[0,1], [1,0]], [3,2]);
x_{[ 0, 1 ]}( 3 ) * x_{[ 1, 0 ]}( 2 )
gap> CanonicalForm(z);
x_{[ 1, 0 ]}( 2 ) * x_{[ 0, 1 ]}( 3 ) * x_{[ 1, 1 ]}( 6 ) * x_{[ 2, 1 ]}( 12 ) * x_{[ 3, 1 ]}( 24 ) * x_{[ 3, 2 ]}( -72 )
```

So if we call the positive roots $r_1, \dots, r_6$, we have $z = x_{r_2}(3)x_{r_1}(2) = x_{r_1}(2)x_{r_2}(3)x_{r_3}(6)x_{r_4}(12)x_{r_5}(24)x_{r_6}(-72)$.

2.3.9 PrintObj (for UnipotChevElem)

▷ `PrintObj(x)`

(method)

▷ `ViewObj(x)`

(method)

Special methods for unipotent elements. (see GAP Reference Manual, section (**Reference: View and Print**) for general information on View and Print). The output depends on the representation of x .

Example

```
gap> Print(x);
UnipotChevElemByRootNumbers( UnipotChevSubGr( "G", 2, Rationals ), \
[ 1 ], [ 2 ] )gap> View(x);
x_{1}( 2 )gap>
```

Example

```
gap> Print(x1);
UnipotChevElemByFundamentalCoeffs( UnipotChevSubGr( "G", 2, Rationals ), \
[ [ 1, 0 ] ], [ 2 ] )gap> View(x1);
x_{[ 1, 0 ]}( 2 )gap>
```

2.3.10 ShallowCopy (for UnipotentChevElem)

▷ ShallowCopy(x)

(method)

This is a special method for unipotent elements.

ShallowCopy creates a copy of x . The returned object is *not identical* to x but it is *equal* to x w.r.t. the equality operator $=$. *Note* that CanonicalForm and Inverse of x (if known) are identical to CanonicalForm and Inverse of the returned object.

(See GAP Reference Manual, section **(Reference: Duplication of Objects)** for further information on copyability)

2.3.11 \= (for UnipotentChevElem)

▷ \=(x, y)

(method)

Special method for unipotent elements. If x and y are identical or are products of the *same* root elements then true is returned. Otherwise CanonicalForm (see CanonicalForm (2.3.8)) of both arguments must be computed (if not already known), which may be expensive. If the canonical form of one of the elements must be calculated and InfoLevel of UnipotentChevInfo is at least 3, the user is notified about this:

Example

```
gap> y := UnipotentChevElemByRN( U_G2, [1,5], [2,7] );
x_{1}( 2 ) * x_{5}( 7 )
gap> z := UnipotentChevElemByRN( U_G2, [5,1], [7,2] );
x_{5}( 7 ) * x_{1}( 2 )
gap> SetInfoLevel( UnipotentChevInfo, 3 );
gap> y=z;
#I CanonicalForm for the 1st argument is not known.
#I           computing it may take a while.
#I CanonicalForm for the 2nd argument is not known.
#I           computing it may take a while.
true
gap> SetInfoLevel( UnipotentChevInfo, 1 );
```

2.3.12 \< (for UnipotentChevElem)

▷ \<(x, y)

(method)

Special method for UnipotentChevElem

This is needed e.g. by AsSSortedList.

The ordering is computed in the following way: Let $x = x_{r_1}(s_1) \dots x_{r_n}(s_n)$ and $y = x_{r_1}(t_1) \dots x_{r_n}(t_n)$, then

$$x < y \Leftrightarrow [s_1, \dots, s_n] < [t_1, \dots, t_n],$$

where the lists are compared lexicographically. e.g. for $x = x_{r_1}(1)x_{r_2}(1) = x_{r_1}(1)x_{r_2}(1)x_{r_3}(0)$ (field elems: [1, 1, 0]) and $y = x_{r_1}(1)x_{r_3}(1) = x_{r_1}(1)x_{r_2}(0)x_{r_3}(1)$ (field elems: [1, 0, 1]) we have $y < x$ (above lists ordered lexicographically).

2.3.13 * (for UnipotChevElem)

▷ *(x, y)

(method)

Special method for unipotent elements. The expressions in the form $x_r(t)x_r(u)$ will be reduced to $x_r(t+u)$ whenever possible.

Example

```
gap> y;z;
x_{1}( 2 ) * x_{5}( 7 )
x_{5}( 7 ) * x_{1}( 2 )
gap> y*z;
x_{1}( 2 ) * x_{5}( 14 ) * x_{1}( 2 )
```

Note: The representation of the product will be always the representation of the first argument.

Example

```
gap> x; x1; x=x1;
x_{1}( 2 )
x_{[ 1, 0 ]}( 2 )
true
gap> x * x1;
x_{1}( 4 )
gap> x1 * x;
x_{[ 1, 0 ]}( 4 )
```

2.3.14 OneOp (for UnipotChevElem)

▷ OneOp(x)

(method)

Special method for unipotent elements. OneOp returns the multiplicative neutral element of x. This is equal to x^0 .

2.3.15 Inverse (for UnipotChevElem)

▷ Inverse(x)

(method)

▷ InverseOp(x)

(method)

Special methods for unipotent elements. We are using the fact

$$\left(x_{r_1}(t_1) \dots x_{r_m}(t_m)\right)^{-1} = x_{r_m}(-t_m) \dots x_{r_1}(-t_1).$$

2.3.16 IsOne

▷ IsOne(x)

(method)

Special method for unipotent elements. Returns true if and only if x is equal to the identity element.

2.3.17 $\wedge$ (powers of UnipotChevElem)

▷ $\wedge(x, i)$ (method)

Integral powers of the unipotent elements are calculated by the default methods installed in GAP. But special (more efficient) methods are installed for root elements and for the identity.

2.3.18 $\wedge$ (conjugation of UnipotChevElem)

▷ $\wedge(x, y)$ (method)

Conjugation of two unipotent elements, i.e. $x^y = y^{-1}xy$. The representation of the result will be the representation of x .

2.3.19 Comm (for UnipotChevElem)

▷ $\text{Comm}(x, y[, \text{"canonical"}])$ (method)

Special methods for unipotent elements.

Comm returns the commutator of x and y , i.e. $x^{-1}y^{-1}xy$. The second variant returns the canonical form of the commutator. In some cases it may be more efficient than $\text{CanonicalForm}(\text{Comm}(x, y))$

2.3.20 IsRootElement

▷ $\text{IsRootElement}(x)$ (property)

IsRootElement returns true if and only if x is a root element, i.e. $x = x_r(t)$ for some root r . We store this property immediately after creating objects.

Note: the canonical form of x may be a root element even if x isn't one.

Example

```
gap> x := UnipotChevElemByRN( U_G2, [1,5,1], [2,7,-2] );
x_{1}( 2 ) * x_{5}( 7 ) * x_{1}( -2 )
gap> IsRootElement(x);
false
gap> CanonicalForm(x); IsRootElement(CanonicalForm(x));
x_{5}( 7 )
true
```

2.3.21 IsCentral (for a unipotent subgroup and element)

▷ $\text{IsCentral}(U, z)$ (method)

Special method for a unipotent subgroup and a unipotent element.

2.4 Symbolic computation

In some cases, calculation with explicit elements is not enough. Unipot provides a way to do symbolic calculations with unipotent elements for this purpose. This is done by using indeterminates (see GAP Reference Manual, (**Reference: Indeterminates**) for more information) over the underlying field instead of the field elements.

Example

```
gap> U_G2 := UnipotChevSubGr("G", 2, Rationals);;
gap> a := Indeterminate( Rationals, "a" );
a
gap> b := Indeterminate( Rationals, "b", [a] );
b
gap> c := Indeterminate( Rationals, "c", [a,b] );
c
gap> x := UnipotChevElemByFC(U_G2, [ [3,1], [1,0], [0,1] ], [a,b,c] );
x_{[ 3, 1 ]}( a ) * x_{[ 1, 0 ]}( b ) * x_{[ 0, 1 ]}( c )
gap> CanonicalForm(x);
x_{[ 1, 0 ]}( b ) * x_{[ 0, 1 ]}( c ) * x_{[ 3, 1 ]}( a ) * x_{[ 3, 2 ]}( a*c )
gap> CanonicalForm(x^-1);
x_{[ 1, 0 ]}( -b ) * x_{[ 0, 1 ]}( -c ) * x_{[ 1, 1 ]}( b*c ) * x_{[ 2, 1 ]}( -b^2*c ) * x_{[ 3, 1 ]}( b^3*c-a ) * x_{[ 3, 2 ]}( b^3*c^2 )
```

References

- [Car89] Roger W. Carter. *Simple groups of Lie type*. Wiley Classics Library. John Wiley & Sons Inc., New York, 1989. Reprint of the 1972 original, A Wiley-Interscience Publication. [3](#), [6](#), [11](#)
- [GAP04] The GAP Group, Aachen, St Andrews. *GAP – Groups, Algorithms, and Programming, Version 4.4*, 2004. [3](#)
- [Hal00] Sergei Haller. Entwicklung und Implementierung eines Algorithmus zur Berechnung von Kommutatoren unipotenter Elemente in Chevalley-Gruppen. Diplomarbeit, Mathematisches Institut, Justus-Liebig-Universität Gießen, April 2000. [3](#)

Index

- *
- for UnipotChevElem, [13](#)
- \=
- for UnipotChevElem, [12](#)
- \~
- conjugation of UnipotChevElem, [14](#)
- powers of UnipotChevElem, [14](#)
- \<
- for UnipotChevElem, [12](#)
- CanonicalForm, [11](#)
- CentralElement, [7](#)
- Comm
- for UnipotChevElem, [14](#)
- GeneratorsOfGroup
- for UnipotChevSubGr, [7](#)
- Inverse
- for UnipotChevElem, [13](#)
- InverseOp
- for UnipotChevElem, [13](#)
- IsCentral
- for a unipotent subgroup and element, [14](#)
- IsOne, [13](#)
- IsRootElement, [14](#)
- IsUnipotChevElem, [7](#)
- IsUnipotChevRepByFundamentalCoeffs, [8](#)
- IsUnipotChevRepByRootNumbers, [8](#)
- IsUnipotChevRepByRoots, [8](#)
- IsUnipotChevSubGr, [5](#)
- NegativeRootsFC, [6](#)
- One
- for UnipotChevSubGr, [6](#)
- OneOp
- for UnipotChevElem, [13](#)
- for UnipotChevSubGr, [6](#)
- PositiveRootsFC, [6](#)
- PrintObj
- for UnipotChevElem, [11](#)
- for UnipotChevSubGr, [5](#)
- Representative, [7](#)
- RootSystem
- for UnipotChevSubGr, [6](#)
- ShallowCopy
- for UnipotChevElem, [12](#)
- Size
- for UnipotChevSubGr, [6](#)
- Unipot, [3](#)
- UnipotChevElemByFC, [9](#)
- for U, root, felem, [9](#)
- UnipotChevElemByFundamentalCoeffs, [9](#)
- element conversion, [10](#)
- for U, root, felem, [9](#)
- UnipotChevElemByR, [9](#)
- for U, root, felem, [10](#)
- UnipotChevElemByRN, [8](#)
- for U, root, felem, [8](#)
- UnipotChevElemByRootNumbers, [8](#)
- element conversion, [10](#)
- for U, root, felem, [8](#)
- UnipotChevElemByRoots, [9](#)
- element conversion, [10](#)
- for U, root, felem, [9](#)
- UnipotChevInfo, [4](#)
- UnipotChevSubGr, [5](#)
- UNIPOT_DEFAULT_REP, [8](#)
- ViewObj
- for UnipotChevElem, [11](#)
- for UnipotChevSubGr, [5](#)